Refining POWGEN data in TOPAS

Preamble

- TOPAS is a diffraction analysis package available in two versions; commercial with a GUI (Bruker-AXS) and academic without a GUI (http://www.topas-academic.net/).
- TOPAS doesn't 'know' about TOF data but may be used in versions 3 and above using the text-based launch-mode. This document assumes version 5 or 6.
- Launch-mode is very powerful, using something close to a computer code to create macros to manipulate/control the refinement in ways the developer never envisaged. Complex restraints, peak profiles, structure solution, rigid bodies, parametric and magnetic refinement are examples of what's possible within version 5. However the jump from the GUI mode in the commercial TOPAS to the more advanced features in launch-mode involves a steep learning curve.
- The TOPAS input file supplied from the POWGEN website is in the form of a template – no direct equivalent of the GSAS *.prm for Fullprof *.irf files exist. The template has been developed with version 5. It should be backwards compatible with version 4.2 and possibly version 3 but has not been tested.
- Template files from different cycles may be used interchangeably as long as correct calibration information for the relevant cycle (zero, difc, difa, alpha0, alpha1, beta0, beta1) is copied across. The template doesn't use the built-in TOF equations as it allows for more flexibility for future developments.
- The template files have been continually developed to improve their layout and add new functionality in the form of different customized macros. The template from 2016-B used here may look different from other cycles.
- Additional datasets such as lab/synchrotron X-ray or single-crystal may be added to these templates. Similarly Pawley, Le Bail and single-peak fitting can be done but the template was set up to simplify structure refinements of POWGEN data as much as possible.
- It is highly recommended that the jEdit text editor be used to work with the input files. Useful features such as color-coding, keyword menus, column editing and text-folding are available when set up correctly
- Information, TOPAS launch-mode tutorials and instructions for setting up jEdit for TOPAS can be found on the following website

  https://community.dur.ac.uk/john.evans/topas_academic/topas_main.htm
- The TOPAS Technical Reference contains examples of the syntax and usage of the native TOPAS keywords/macros
- The TOPAS Wiki is a useful source of information, user-supplied macros and a user forum

  http://topas.dur.ac.uk/topaswiki/doku.php

2016-B TOPAS Template structure – top half



2016-B TOPAS Template structure – bottom half



jEdit – folding

- Folding is a very useful feature, sections can be hidden away to simplify navigation of long text files.
- A fold is started by typing ‘{{{ and finished by ‘}}}

- The text folding is not setup in the instructions given by John Evans’ website but is easily set in the jEdit Global Options as shown by changing the Folding mode to ‘explicit’





Refinements – initial setup

- Expand the section to add the required filenames

```
'{{{ select frames to use here and enter filenames

#define USE_FRAME1  macro filename_frame1 {PG3_?-1.xye}  macro start_end_frame1 {start_X 4800  finish_X 70000}     '0.533 CWL
#define USE_FRAME2  macro filename_frame2 {PG3_?-2.xye}  macro start_end_frame2 {start_X 6775  finish_X 104000}     '1.066 CWL
#define USE_FRAME3  macro filename_frame3 {PG3_?-3.xye}  macro start_end_frame3 {start_X 9930 finish_X 121700}    '1.333 CWL
#define USE_FRAME4  macro filename_frame4 {PG3_?-4.xye}  macro start_end_frame4 {start_X 26501 finish_X 207600}     '2.665 CWL
#define USE_FRAME5  macro filename_frame5 {PG3_?-5.xye}  macro start_end_frame5 {start_X 38400 finish_X 280000}     '3.731  CWL
#define USE_FRAME6  macro filename_frame6 {PG3_?-6.xye}  macro start_end_frame6 {start_X 51340 finish_X 346000}     '4.797  CWL

'}}}
```

Filenames, start and finish of data to fit in TOF for each frame

- Frames not required may either be commented out or deleted

```
'{{{ select frames to use here and enter filenames

'#define USE_FRAME1  macro filename_frame1 {PG3_?-1.xye}  macro start_end_frame1 {start_X 4800  finish_X 70000}     '0.533 CWL
#define USE_FRAME2  macro filename_frame2 {PG3_29601-2.xye}  macro start_end_frame2 {start_X 6775  finish_X 104000}     '1.066 CWL
#define USE_FRAME4  macro filename_frame4 {PG3_29603-4.xye}  macro start_end_frame4 {start_X 26501 finish_X 207600}     '2.665 CWL

'}}}
```

```
'{{{ r-factors and control information
r_exp                0.939678895
r_exp_dash           3.33349497
r_wp                 4.30358706
r_wp_dash            15.2669023
r_p                  5.54086233
r_p_dash             26.4086192
gof                  4.57984859
weighted_Durbin_Watson  1.1501352
'continue_after_convergence
do_errors
```

Control information includes whether to calculate errors

Refinements – initial setup

- Sections relating unused frames may be deleted but will not cause an error when the refinement is run
- Expand a section in use, check parameters and add label to phase. Add 'view_structure' under the 'str' if you wish to see the structure during refinement



- Additional phases may be added by inserting another 'str'. Unless you want them to be constrained change the variable names for the scale, etc.

Adding structure information

- Structure information may either be typed in or imported from a CIF file. Beware – the jEdit CIF import occasionally needs two attempts to get the correct CIF file – make sure it's the right one



Some spacegroups have options for different origins (e.g. Fd-3m) but CIF files often don't specify which they used – CHECK!



- Correct any oddities from the CIF file, e.g. oxidation states, oxygens in water denoted by 'W', etc.
- Any special positions with recurring decimals much be declared as a fraction, e.g. "x 0.3333" should be "x = 1/3;" The ";" is important and must appear after any equation – without it TOPAS will probably throw an error but not always in the location where you expect!



X remove the 'str' statement

"for xdds {}" and "for strs {}" loops tie together shared parameters from different datasets/structures.
IMPORTANT – DO NOT use "@" to set parameters to refine inside "for xdds {}" or "for strs {}" as it will create independent variables for each dataset and not do what you expect. Make up specific variables names, e.g. b_o, to ensure you get what you want.



a parameter given variable name. "!" means it's fixed

Lattice parameters tied together with an equation.

TOPAS uses B_{iso} – not U_{iso}

"num_posns 0" will populate with site multiplicity after refinement

SRM-660b contains boron-11 isotope

Rerun refinement after convergence – if the real-time R_{wp} loops around each time around then probable that a '@' is in there somewhere..

Run the refinements…

Run refinement to completion

Run refinement step-wise



Turns animation on and off. Affects both difference plot and image of structure

If selected will display graphical representation of the correlation matrix (v5 and above)



Parameters pane…

- The parameters pane (F2 key or wrench button in interface) may be used to choose which dataset(s) to see in the main window.

- In the GUI mode this the main route to control and interact with the refinement, e.g. instrumental parameters, structure, microstructure, Pawley refinements….
- In launch-mode it is mostly inactive – the only useful pane under structures is the view of the generated hkls under the structure. You may also customize the display of the difference plot in the parameters pane.



	h	k	l	m	d	th2	F^2
1	10	9	3	48	0.30157	6801.61572	9958.742
2	10	8	5	48	0.30237	6819.60059	4492.188
3	12	6	3	48	0.30237	6819.60059	109.143
4	11	8	2	48	0.30237	6819.60059	14601.884
5	13	4	2	48	0.30237	6819.60059	171.550
6	9	9	5	24	0.30398	6856.00195	4938.615
7	13	3	3	24	0.30398	6856.00195	3765.112
8	11	8	1	48	0.30480	6874.42236	1751.721
9	11	7	4	48	0.30480	6874.42236	3818.339
10	13	4	1	48	0.30480	6874.42236	4529.160
11	10	7	6	48	0.30562	6892.99219	4973.012
12	0	13	4	24	0.30562	6892.99219	2528.995
13	10	9	2	48	0.30562	6892.99219	1026.203
14	12	5	4	48	0.30562	6892.99219	9672.438
15	0	11	8	24	0.30562	6892.99219	741.427
16	12	6	2	48	0.30645	6911.71289	821.220
17	11	6	5	48	0.30813	6949.61670	372.317
18	10	9	1	48	0.30813	6949.61670	2452.741

Peak profiles…

- There are different options for fitting peak profiles in TOPAS. All of them assume a particular dependence with d-spacing. The two built-in functions are TOF_CS_L and TOF_CS_G.

  $FWHM_{lor} = 0.1 \times \text{difc} \times \text{d-spacing}^2 / \text{variable}_{TOF\_CS\_L}$

  $FWHM_{gauss} = 0.1 \times \text{difc} \times \text{d-spacing}^2 / \text{variable}_{TOF\_CS\_G}$

- An alternative peak broadening function (written by John Evans) is supplied in the template that frequently does a better job of fitting peak profiles of POWGEN data. This is a macro called tof_sample_peakshape that is declared in the template itself that applies a pseudo-Voigt as follows

  $FWHM_{lor} = \text{lor} (\text{dsp} \times \text{d-spacing} + \text{dspsq} \times \text{d-spacing}^2)$

  $FWHM_{gauss} = 1\text{-lor} (\text{dsp} \times \text{d-spacing} + \text{dspsq} \times \text{d-spacing}^2)$

- A macro called POWGEN_instrument is also described in the template TOF macros that follows uses GSAS-style sigmas and gammas. The sigma and gamma variables are not declared in the template as-supplied under the datasets but may be added by copy/paste with POWGEN_instrument to call the macro.

  $FWHM_{lor} = \sqrt{(\text{gam0} + \text{gam1} \times \text{d-spacing} + \text{gam2} \times \text{d-spacing}^2)^2}$

  $FWHM_{gauss} = \sqrt{(\text{sig0} + \text{sig1} \times \text{d-spacing}^2 + \text{sig2} \times \text{d-spacing}^4)^2}$

- Results of each profile function will vary depending on the particular sample
- The Stephens model for anisotropic strain broadening is either built-in (V5 and above) or available via the TOPAS wiki (V4.2 and below)
- More sophisticated models exist (see TOPAS wiki) but are frequently written for monochromatic radiation and may require modification for TOF

User-defined backgrounds

- The default Chebychev polynomial background function is frequently inadequate for high-Q TOF data due to oscillations caused by diffuse scattering.

- In V5 and above it is possible to write user-defined background functions via 'fit objects'.
- These may be written in TOF (X) or Q which may be defined for a dataset as

```
local Q = 2 Pi Constant(difc) / (X – Constant(zero));
```

- One example of a user-defined background would be the diffuse scattering from positional disorder as described in the GSAS manual
- This has 3 refinable variables we'll call ba2, ba3 and sc1
- The fit object would be calculated as

```
prm ba2  v1
prm ba3  v2
prm sc1   v3
Fit_obj !v1 = sc1 (Sin(Q ba2)/Q ba2) Exp(-0.5 ba3 Q^2);
```

- The fit object can then be plotted using

```
Plot_Fit_Obj(v1, "Fit Obj1")
```

Combined Neutron/X-ray refinements

- A space has been left in the template to allow the addition of another dataset, e.g. lab XRD or synchrotron dataset.
- The standard method of adding a XRD/constant wavelength dataset should be followed using a similar approach to the TOF data.
- xdd filename.xxx loads the data
- bkg uses a Chebychev background
- lam is the wavelength
- Lorentz polarization LP_factor()
- zero_error for zero point correction
- XRD peakshape functions such as CS_L() should be used instead of TOF functions
- Should you wish to add a dataset from another TOF instrument (e.g. NOMAD) please contact one of the instrument team to correctly format the template

```
'{{{ information specific to frame 6 refinement here  4.797
#ifdef USE_FRAME6
TOF_XYE(filename_frame6, = Yobs_dx_at(Xo) .5;) start_end_frame6
r_wp 0 r_exp 0 r_p 0 r_wp_dash 0 r_p_dash 0 r_exp_dash 0 weighted_Durbin_Watson 0 gof 0  'frame 6
bkg @  0 0 0 0 0

local !zero   -53.70751         'don't refine
local !dif_c  22600.51807       'don't refine
local !alpha0 -530.46172        'don't refine
local !alphal 1219.01128       'don't refine
local !beta0  491.02832         'don't refine
local !betal  -4362.10988        'don't refine

local !dif_a  0   'refine if necessary

local !mu  0 min 0     'refine if necessary
TOF_misc
str
   phase_name ?
   r_bragg  0
   scale scal_f6  1
    peak_type pv
     TOF_CS_L(!cslf6, 580.50382, dif_c)
     TOF_CS_G(!csgf6, 259.04919, dif_c)
     tof_sample_peakshape(!lor_f6, 6.76091017e-008,!dsp_f6, 1.00000,!dspsq_f6, 2.36777)

#endif
'}}}

' add information for additional dataset here, e.g. lab XRD, 11BM, etc
'xdd .....     ' import file
' bkg ...
' lam ...      'instrument info
' str ..       'dataset-specific structure info (scale, broadening, etc)
```

More advanced refinement.. ADPs

- As a refinement progresses different variables will need to be turned on and off with more sophisticated models applied.
- ADPs are used infrequently with powder X-ray data but are common with TOF neutron data.
- TOPAS is unusual in that it uses Bs for isotropic displacement parameters and Us for anisotropic displacement parameters (ADPs).
  - The relationship between B_{iso} and U_{iso} is: $U_{iso} = B_{iso}/8\pi^2$
- The correct syntax for refining ADPs may be created by running the macro adps for each required site.

- **Beware**
  - A bug crept into some versions of TOPAS where the correct site symmetry constraints for ADPs were not always created for some space groups; check Chapter 8.3 in Volume C of the International Tables if in doubt
  - The adps macro places the '@' refinement flag before the parameters which cannot be used safely inside a 'for xdds' loop. Change the '@' for a unique variable name such as 'u11_o1'.

ADPs and constraints

- Example is a refinement of corundum with 2016-B POWGEN data
- Unique variable names used for each refined U_{aniso}
- Constraints use the keyword Get() in equations to link parameters for the same site, e.g. in the oxygen site

  o_u12 = Get(u22)/2;
- This gets the value of u22 for oxygen without specifying the variable name

```
for xdds {
 for strs 1 to 1 {
     r_bragg  0.670561565
     phase_name corundum
     space_group R-3c
     Trigonal(a_corundumori  4.760054`_0.000027 min =4.7400; max =4.7800;,c_corundumori  12.993475`_0.000087 min =12.9000; max =13.1000;)
     site Al1 num_posns  12 x =0; :  0.00000 y =0; :  0.00000 z z1_corundumori  0.35213`_0.00003 occ Al+3  1
     ADPs { al_u11  0.00273`_0.00009 = Get(u11); al_u33  0.00299`_0.00012 al_u12  =Get(u11)/2;:0.00136`_0.00005   0   0 }
     site O2 num_posns  18 x x2_corundumori  0.69363`_0.00003 y =0; :  0.00000 z =1/4; :  0.25000 occ O-2  1
     ADPs { o_u11  0.00307`_0.00008 o_u22  0.00337`_0.00008 o_u33  0.00329`_0.00007 o_u12  =Get(u22)/2;:0.00169`_0.00004 o_u13  0.00025`_0.00003 =Get(u13)*2;:0.00050`_0.00005 }


}
```

- Symmetry check for R-3c. International Tables Volume C. Chapter 8.3 (uses β_{aniso} but same relationships)

  Al = 12c with site symmetry '3.' $\beta_{11} = \beta_{22} = 2\beta_{12}$; $\beta_{13} = \beta_{23} = 0$

  O = 18e with site symmetry '.2' $\beta_{22} = 2\beta_{12}$; $2\beta_{13} = \beta_{23}$



	z (Al)	U11 (Al)	U33 (Al)	x (O)	U11 (O)	U22 (O)	U33(O)	U13 (O)
Laue single crystal neutron (GSAS)	0.35216(2)	0.00282(5)	0.00287(7)	0.30635(3)	0.00292(4)	0.00326(5)	0.00330(4)	
POWGEN (2016B)	0.35213(3)	0.00273(9)	0.00299(12)	0.30637(2)	0.00307(2)	0.00337(8)	0.00329(7)	0.00025(3)

Outputting data for publication…

- TOPAS 3/4 have very basic CIF output (Out_CIF_STR) but versions 5 and 6 have more sophisticated output built in for the structure ADPs (Out_CIF_ADPs), bond-lengths/angles (Out_CIF_Bonds_Angles), etc.
- Additional macros may be written to output the fit data in a format suitable for external plotting/publication – many can be found on the TOPAS Wiki.
- In version 5 and above fit data may be extracted from the parameters pane using "Save if displayed Yobs, Ycalc, Diff, Phases, Bkg"





Pawley/Le Bail refinements

- Structureless cell refinements (Pawley and Le Bail) are easy to perform with TOPAS but the template was designed for Rietveld refinements and may require some modification.
- Pawley/Lebail fits are triggered by using the keyword hkl_Is instead of str. Under for xdds {} the for strs {} will need changing to for hklis {}
- TOPAS will append the hkl list underneath the keyword space_group
- If space_group appears under for xdds {} and multiple frames are being fitted then the list from only the final dataset will be output.
  - In this instance move the space_group declaration to underneath each hkl_Is phase.
  - Remove or comment out redundant keywords such as the scale factor.
  - Fix absorption (mu) at zero
  - Remove or comment out all site lines under hkl or for hklis {}
- Pawley/Le Bail and Rietveld refinements can be combined in a single refinement but care must be taken to ensure everything is in the correct order under for xdds {}

- Note: POWGEN data can contain a very large number of reflections. Pawley refinements have individual position and intensity values for each reflection
  - The correlation matrix becomes extremely large so calculating errors with do_errors may take a very long time and use a lot of computing resources.

Pawley/Le Bail refinements

- Modified text required to switch to a Pawley refinement.
- To do a Le Bail refinement add the keyword

  lebail 1

  under each hkl_Is

```
'{{{ information specific to frame 6 refinement here  4.797
#ifdef USE_FRAME6
TOF_XYE(filename_frame6, = Yobs_dx_at(Xo) .5;) start_end_frame6
r_wp 0 r_exp 0 r_p 0 r_wp_dash 0 r_p_dash 0 r_exp_dash 0 weighted_Durbin_Watson 0 gof 0  'frame 6
bkg @  0 0 0 0 0

local !zero   -53.70751         'don't refine
local !dif_c  22600.51807      'don't refine
local !alpha0 -530.46172       'don't refine
local !alpha1 1219.01128      'don't refine
local !beta0  491.02832        'don't refine
local !beta1  -4362.10988       'don't refine

local !dif_a  0   'refine if necessary

local !mu  0 min 0     'refine if necessary
TOF_misc
hkl_Is
   phase_name ?
   r_bragg  0
  ' scale scal_f6  1
   peak_type pv
    TOF_CS_L(!cslf6, 580.50382, dif_c)
    TOF_CS_G(!csgf6, 259.04919, dif_c)
    tof_sample_peakshape(!lor_f6, 6.76091017e-008,!dsp_f6, 1.00000,!dspsq_f6, 2.36777)
    space_group "?"
#endif
'}}}

' add information for additional dataset here, e.g. lab XRD, 11BM, etc
'xdd .....     ' import file
' bkg ...
' lam ...      'instrument info
' str ..       'dataset-specific structure info (scale, broadening, etc)

'-------------------------------------------------------------

'{{ overall structural information here

'-------------------------------------------------------------
'Put overall structural and peak shape information here
'Put the overall peak shape here so it can be refined for all banks in calibration then fixed for all banks
'-------------------------------------------------------------
for xdds {
 for hklis 1 to 1 {
  ' space_group "?"
   phase_name ?
   a ?
   b ?
   c ?
   al ?
   be ?
   ga ?
   'site  ?1 x     ? y ? z ?    occ ?  1.0  beq 0.3

}

' add second phase here if needed for strs 2 to 2 {}  also needs second str info under each dataset used
}
'}}}
```